

Bäcklund Transformation

Graham W. Griffiths¹

City University, UK.

oo0oo

Contents

1	Introduction	1
2	Applications of the Bäcklund transformation method	2
2.1	Bäcklund transformation of Burgers equation	2
2.2	Auto-Bäcklund transformation of the sine-Gordon equation	6
2.3	Miura transformation of the KdV equation	8
2.4	Bäcklund transformation of the KdV equation	10
3	Nonlinear superposition	13
3.1	Bianchi's Permutability Theorem	13
3.2	Nonlinear superposition applied to the KdV equation	14
3.3	Three-soliton solution to the KdV equation using nonlinear superposition	17
3.4	Nonlinear superposition applied to the sine-Gordon equation	20

1 Introduction

The *Bäcklund transformation* (BT) was introduced in 1880 by A. V. Bäcklund [Bac-80] as a transformation between 3-dimensional surfaces. By applying Bianchi's integrability condition,

$$\frac{\partial^2 u}{\partial x_1 \partial x_2} - \frac{\partial^2 u}{\partial x_2 \partial x_1} = 0, \quad (1)$$

discussed subsequently in section (3), a Bäcklund transformation can be used to generate a series of solutions to a partial differential equation from a known trivial solution.

A Bäcklund transformation transforms a nonlinear partial differential equation into another partial differential equation. Thus, a solution to the second partial differential equation must be compatible to the first partial differential equation. Hence, application of the Bäcklund transformation can provide a powerful method for generating solutions to nonlinear PDEs.

Examples of well known Bäcklund transformations are [Miu-68a][Miu-68b] a) the *Miura transformation*, which transforms the KdV equation into the modified KdV equation and, b) the *Gardner transformation* which transforms the KdV equation into the *Gardner equation*.

An *auto-Bäcklund transformation* is a transform that leaves a partial differential equation invariant. It can also provide a method whereby a known solution to a partial differential equation can be transformed into a different second solution to the same partial differential equation. This solution can then be used to obtain a third solution, and so on. However, it may not be possible to carry this process on to find an infinite sequence, or even many solutions, as they can repeat at some stage.

There is no known general systematic method for generating Bäcklund transformations, and therefore this subject continues to be an active area of research. Detailed discussion on the origin of the Bäcklund transformation, along with a wide variety of applications,

¹email: graham@griffiths1.com

can be found in the excellent books by Rogers and Shadwick [Rog-82], and Dodd, et al. [Dod-82].

2 Applications of the Bäcklund transformation method

2.1 Bäcklund transformation of Burgers equation

Burgers equation is given by,

$$u_t + uu_x - \alpha u_{xx} = 0, \quad (2)$$

where $u = u(x, t)$ and α is the conduction coefficient, and we wish to find a solution using the Bäcklund transformation method.

A Bäcklund transformation that transforms Burgers equation to the *heat equation*,

$$v_t = v_{xx}, \quad (3)$$

is given by

$$v_x = -\frac{vu}{2\alpha}, \quad v_t = \frac{u^2v}{4\alpha} - \frac{vu_x}{2}, \quad (4)$$

where $v = v(x, t)$ and the first equation represents the x -part of the BT and the second equation the t -part.

That eqn. (4) is actually the required BT can be easily demonstrated. First we differentiate the x -part with respect to t and the t -part with respect to x , then on subtracting we obtain after some manipulation,

$$u_t + uu_x - \alpha u_{xx} = -\frac{uv_t}{v} - \frac{u^2v_x}{2v} + \frac{\alpha v_x u_x}{v}. \quad (5)$$

Now, the left hand side of eqn. (5) is the Burgers equation (2) and is therefore equal to zero. Consequently, for compatibility, the right hand side must also be equal to zero. Therefore, on setting the right hand side to zero and rearranging, we obtain

$$v_t = -\frac{v}{u} \left(\frac{u^2v_x}{2v} - \frac{\alpha v_x u_x}{v} \right). \quad (6)$$

We now subtract α multiplied by the derivative of the x -part of eqn. (4) from eqn. (6), to obtain

$$v_t - \alpha v_{xx} = \frac{u_x}{2u} (2\alpha v_x + vu). \quad (7)$$

But the left hand side of eqn. (7) is the heat equation (3), and is therefore equal to zero. Assuming that $u \neq 0$, compatibility considerations require that the bracketed term on the right hand side is equal to zero. Therefore, on setting the bracketed term to zero and rearranging we arrive at the x -part of the Bäcklund transformation, eqn. (4), i.e.

$$v_x = -\frac{vu}{2\alpha}. \quad (8)$$

Therefore, by definition we have $(2\alpha v_x + vu) = 0$. Thus, we have arrived at the point where, given a solution to the heat eqn (3), we can derive a solution to Burgers equation by employing the Bäcklund transformation, eqn. (4). A maple program that derives this result is given in Listing (1).

Listing 1: Maple program that performs a Bäcklund transformation of a *heat equation* solution 1 to a *Burgers equation* solution

```

# Backlund transformation of Heat equation solution
# to Burgers equation solution
restart; with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t)): alias(v=v(x,t)):
pde_HeatEqn:=diff(v,t)-alpha*diff(v,x,x)=0;
pde_Burgers:=diff(u,t)+u*diff(u,x)-alpha*diff(u,x,x)=0;
BTx:=diff(v,x)=-1/(2*alpha)*u*v;
BTt:=diff(v,t)=1/(4*alpha)*u^2*v - v*diff(u,x)/2;
eqn1:=diff(BTx,t)-diff(BTt,x);
eqn2:=expand(eqn1*4*alpha);
# Rearrange eqn2
eqn3:=-remove(has,rhs(eqn2),{diff(v,x),diff(v,t)}) =
select(has,rhs(eqn2),{diff(v,x),diff(v,t)});
# Simplify eqn3
eqn4:=expand(eqn3/(2*v));
# LHS of eqn4 is Burgers eqn and, thus, equal to zero! Therefore,
eqn5:=isolate(eval(rhs(eqn4)*(2*v)=0),diff(v,t));
eqn6:=alpha*diff(BTx,x);
eqn7:=expand(eqn5-eqn6);
# LHS eqn7 is Heat eqn and thus equal to zero.
# Therefore, compatibility considerations require that
# the RHS of eqn7 is also equal to zero!
eqn8:=simplify(eval(rhs(eqn7)=0),symbolic);
# For eqn8 to be true, the bracketed term must equal zero.
#This is true as it is equal to BTx, which is zero by definition, i.e.
eqn9:=isolate(eqn8, diff(v,x));

```

As an example, we consider the following classical source solution to to the heat equation(3),

$$v = \frac{1}{\sqrt{(4\alpha\pi t)}} \exp(-x^2/4\alpha t), \quad t > 0, \quad (9)$$

which, from the x -part of the Bäcklund transformation, $u = 2\alpha v_x/v$, we obtain the the following solution to Burgers equation (2),

$$u = \frac{x}{t}. \quad (10)$$

A plot of this solution if given in Fig (1).

A second example is a slightly different source solution to to the heat equation(3),

$$v = 1 + \frac{1}{\sqrt{(4\alpha\pi t)}} \exp(-x^2/4\alpha t), \quad t > 0, \quad (11)$$

which, from the x -part of the Bäcklund transformation, $u = 2\alpha v_x/v$, we obtain the following very different solution to Burgers equation (2),

$$u = \frac{x}{t} \frac{\frac{1}{\sqrt{(4\alpha\pi t)}} \exp(-x^2/4\alpha t)}{1 + \frac{1}{\sqrt{(4\alpha\pi t)}} \exp(-x^2/4\alpha t)}. \quad (12)$$

A plot of this solution if given in Fig (2).

A third (variable separable) solution to the heat equation (3) is,

$$v = A \sin\left(\frac{\pi x}{L}\right) \exp\left(-\frac{\alpha\pi^2}{L^2}t\right), \quad (13)$$




Figure 1: Initial profile at $t = 0.1$ of Burgers equation solution, $u(x, t) = x/t$. The solution moves towards a horizontal profile such that $u(x, \infty) = 0$.

where A and L are arbitrary constants. Thus, again from $u = 2\alpha v_x/v$ we obtain the following stationary (non-time-varying) solution to Burgers equation (2),

$$u = -\frac{2\alpha\pi}{L} \cot\left(\frac{\pi x}{L}\right). \quad (14)$$

A plot of this solution is given in Fig (3).

Thus, the above examples demonstrate that the Bäcklund transformation can produce a variety of very different solutions to the Burgers equation.

A Maple program that derives these results and generates accompanying animations is given in Listing (2).

Listing 2: Maple program that performs a Bäcklund transformation of a *heat equation* solution 2 to a *Burgers equation* solution

```
# Backlund transformation of Heat equation solution 2
# to Burgers equation solution
restart;with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t)): alias(v=v(x,t)):
# Heat Eqn
pde_HeatEqn:=diff(v,t)-alpha*diff(v,x,x)=0;
# Burgers Eqn
pde_BurgersEqn:=diff(u,t)+u*diff(u,x)-alpha*diff(u,x,x)=0;
# Known Heat eqn solution (1) - point heat source
eqn1:=v=1/sqrt(4*alpha*Pi*t)*exp(-x^2/(4*alpha*t));
#assign(eqn1);
# Check Heat eqn solution (1)
pdetest(eqn1,pde_HeatEqn);v1:=rhs(eqn1);
# Generate Burgers eqn solution (1)
eqn2:=u=-2*alpha*diff(v1,x)/v1;
# Check Burgers eqn solution (1)
pdetest(eqn2,pde_BurgersEqn);
# Burgers equation plot data
dat1:={alpha=1};
# Plot results
# =====
z1:=subs(dat1,rhs(eqn2)); # Set data values
# Animate solution
animate(z1,x=-10..10,t=0.1..5,
```




Figure 2: Initial profile at $t = 0$ of Burgers equation *N-wave* solution, $u(x, t) = \frac{x}{t} \frac{1}{\sqrt{(4\alpha\pi t)}} \exp(-x^2/4\alpha t) / \left(1 + \frac{1}{\sqrt{(4\alpha\pi t)}} \exp(-x^2/4\alpha t)\right)$ for $\alpha = 1$. The solution flattens out to a horizontal profile such that $u(x, \infty) = 0$.




Figure 3: Initial profile at $t = 0.1$ of Burgers equation *stationary* solution, $u(x, t) = -\frac{2\alpha\pi}{L} \cot\left(\frac{\pi x}{L}\right)$ for $\alpha = 1$ and $L = 100$.

```

numpoints=300,frames=50, axes=framed,
labels=["x","u"],thickness=3,
title="Burgers equation solution",
labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
titlefont=[TIMES, ROMAN, 16]);
# Known Heat eqn solution (2) - point heat source
eqn3:=v=1+1/sqrt(4*alpha*Pi*t)*exp(-x^2/(4*alpha*t));
# Check Heat eqn solution (2)
pdetest(eqn3,pde_HeatEqn);v2:=rhs(eqn3);
# Generate Burgers eqn solution (2)
eqn4:=u=-2*alpha*diff(v2,x)/v2;
# Check Burgers eqn solution (2)
pdetest(eqn4,pde_BurgersEqn);
# Burgers equation plot data
dat2:={alpha=1};
# Plot results
# =====

```

```

z2:=subs(dat2,rhs(eqn4)); # Set data values
# Animate solution
animate(z2,x=-2..2,t=0.1..5,
  numpoints=300,frames=50, axes=framed,
  labels=["x","u"],thickness=3,
  title="Burgers equation - N-wave solution",
  labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
  titlefont=[TIMES, ROMAN, 16]);
# Known Heat eqn solution (3)
eqn5:=v=A*sin(Pi*x/L)*exp(-alpha*(Pi/L)^2*t);
# Check Heat eqn solution (3)
pdetest(eqn5,pde_HeatEqn);v3:=rhs(eqn5);
# Generate Burgers eqn solution (3)
eqn6:=u=-2*alpha*diff(v3,x)/v3;
# Check Burgers eqn solution (3)
pdetest(eqn6,pde_BurgersEqn);
# Burgers equation plot data
dat3:={alpha=1,A=1,L=100};
# Plot results
# =====
z3:=subs(dat3,rhs(eqn6)); # Set data values
# Animate solution
animate(z3,x=0.1..10,t=0..80,
  numpoints=300,frames=50, axes=framed,
  labels=["x","u"],thickness=3,
  title="Burgers equation - stationary solution",
  labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
  titlefont=[TIMES, ROMAN, 16]);

```

2.2 Auto-Bäcklund transformation of the sine-Gordon equation

The *sine-Gordon* equation is given by

$$u_{xt} = \sin u, \quad (15)$$

and we wish to find a solution using the Bäcklund transformation method. An auto-Bäcklund transformation for eqn. (15) is defined by the following pair of coupled partial differential equations.

$$\begin{aligned} v_x &= u_x + 2\lambda \sin\left(\frac{v+u}{2}\right), \\ v_t &= -u_t + \frac{2}{\lambda} \sin\left(\frac{v-u}{2}\right), \end{aligned} \quad (16)$$

where $u = u(x, t)$ and $v = v(x, t)$. We can easily check that this is so by differentiating the first equation by t and the second by x to obtain,

$$\begin{aligned} v_{xt} &= u_{xt} + 2 \sin\left(\frac{v-u}{2}\right) \cos\left(\frac{v+u}{2}\right), \\ v_{xt} &= -u_{xt} + 2 \sin\left(\frac{v+u}{2}\right) \cos\left(\frac{v-u}{2}\right). \end{aligned} \quad (17)$$

We see immediately that this is correct by equating eqns. (17) when we obtain the original eqn. (15), whilst adding them yields a second sine-Gordon equation, i.e.

$$v_{xt} = \sin v. \quad (18)$$

This means that from a known solution to eqn. (15), we should be able to use the transformation of eqns. (16) to obtain a second solution.

Clearly, $u = 0$ is a solution to eqn. (15) and, on substituting this solution ² is into the auto- Bäcklund transformation of eqn. (16), we obtain,

$$v_x = 2\lambda \sin \frac{v}{2}, \quad v_t = \frac{2}{\lambda} \sin \frac{v}{2}, \quad (19)$$

which can be readily solved for v . Integrating both equations we obtain,

$$\int \frac{dv}{\sin \frac{v}{2}} = 2\lambda \int dx, \quad \rightarrow \quad 2\lambda x = 2 \ln (\csc (1/2v) - \cot (1/2v)) + f(t), \quad (20)$$

$$\int \frac{dv}{\sin \frac{v}{2}} = \frac{2}{\lambda} \int dt, \quad \rightarrow \quad 2\frac{t}{\lambda} = 2 \ln (\csc (1/2v) - \cot (1/2v)) + g(x). \quad (21)$$

Then, after subtracting and rearranging we obtain,

$$f(t) - 2\lambda x = g(x) - 2\frac{t}{\lambda} \quad (22)$$

from which it follows that, for compatibility, we must have,

$$f(t) = K - \frac{2t}{\lambda}, \quad g(x) = K - 2\lambda x, \quad (23)$$

where K is an arbitrary constant. Thus, on adding eqns. (20) and (21) and rearranging, we obtain the following new *kink soliton* solution to the sine-Gordon equation,

$$\tan \frac{v}{4} = C \exp \left(\lambda x + \frac{t}{\lambda} \right), \quad (24)$$

where $C = \exp(-K/2)$.

A plot of this solution is given in Fig (4).




Figure 4: Initial profile at $t = -20$ of the sine-Gordon equation *kink soliton* solution, $v(x, t) = 4 \arctan \left[C \exp \left(\lambda x + \frac{t}{\lambda} \right) \right]$, for $C = 1$ and $\lambda = 1$. Solution move right to left.

In theory we can now use this solution to obtain a second analytical solution, and the second to obtain a third, etc. However, the resulting equations are not easily solved, even with a CAS such as Maple. Nevertheless, we can use this result, along with *nonlinear*

²This starting point is known as the *initiating* or *seed* solution

superposition, to readily obtain new solutions. This approach is demonstrated in a later section.

A maple program that derives these results and generates an accompanying animation is given in Listing (3).

Listing 3: Maple program that derives a solution to the Sine-Gordon equation using the Auto-Bäcklund transformation

```
# Derivation of solution to Sine-Gordon Equation using
# Auto-Backlund Transformation.
restart;with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t)): alias(v=v(x,t)):
# Auto-Backlund transformation for sine-Gordon equation
aBT[1]:= (u,v) -> diff(v,x)=diff(u,x)+2*lambda*sin((v+u)/2);
aBT[2]:= (u,v) -> -diff(v,t)=diff(u,x)+(2/lambda)*sin((v-u)/2);
# Set u=0 in aBT
ode1:=aBT[1](0,v); ode2:=aBT[2](0,v);
# Negate alias'
alias(u=u): alias(v=v):
# Integrating ode1 and ode2
eqn1:=2*lambda*x=int(1/sin(v/2),v)+f(t);
eqn2:=(2/lambda)*t=int(1/sin(v/2),v)+g(x);
# rearranging: eqn3:=eqn1-eqn2;
eqn3:=eqn1-eqn2;
#eqn3:=2*lambda*x+f(x) = g(t)+2*x/lambda;
# Therefore we must have:
f(t):=K-2*t/lambda; g(x):=K-2*lambda*x;
eqn4:=eqn1+eqn2;
eqn5:=isolate(eqn4, ln(csc(1/2*v)-cot(1/2*v)));
eqn6:=exp(lhs(eqn5))=exp(rhs(eqn5));
eqn7:=simplify(subs(K=-2*ln(C),eqn6), power);
eqn8:=simplify(convert(eqn7,tan),symbolic);
eqn9:=convert(eqn8,tan);
# Sine-Gordon equation plot data
dat1:={C=1,lambda=1};
# Plot results
# =====
z1:=subs(dat1,rhs(isolate(eqn9,v))); # Set data values
# Animate solution
animate(z1,x=-30..30,t=-20..20,
  numpoints=300,frames=50, axes=framed,
  labels=["x","u"],thickness=3,
  title="Sine-Gordon equation solution",
  labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
  titlefont=[TIMES, ROMAN, 16]);
```

2.3 Miura transformation of the KdV equation

The *Korteweg-de Vries* (KdV) equation is defined as,

$$u_t - 6uu_x + u_{xxx} = 0, \quad (25)$$

and we desire to obtain a solution to this equation by application of the *Miura transformation* [Miu-68a], given by

$$u = v^2 + v_x, \quad (26)$$

where $u = u(x, t)$ and $v = v(x, t)$. This equation, when substituted directly into eqn. (25), yields

$$\left(2v + \frac{\partial}{\partial x}\right) (v_t - 6v^2v_x + v_{xxx}) = 0. \quad (27)$$

Thus, if v is a solution to,

$$v_t - 6v^2v_x + v_{xxx} = 0, \quad (28)$$

which is known as the *modified KdV* (mKdV) equation, then eqn. (4) will yield a solution to eqn. (25). However, a solution to eqn. (25) does not imply a solution to equation(28) owing to the additional operator of eqn. (27).

Thus, equations (4) and (28) can be regarded as a Bäcklund transformation as they provide a derivation route to obtain solutions to the KdV equation. For example, a *kink* soliton solution to eqn. (28) is,

$$v = -k \tanh [k (x - x_0 - 2k^2t)]; \quad (29)$$

and, on substituting eqn. (29) into eqn. (4), we obtain the following *hump* soliton solution to eqn. (25)

$$u = k^2 \left(2 \tanh [k (x + x_0 + 2k^2t)]^2 - 1 \right). \quad (30)$$

Plots of these solutions are given in Figs (5) and (6).




Figure 5: Initial profile at $t = -50$ of mKdV equation *anti-kink soliton* solution, $v(x, t) = -k \tanh [k (x - x_0 - 2k^2t)]$ for $k = 1$ and $x_0 = 0$. Solution move right to left.

A Maple program that generate a new KdV solution from a known mKdV solution by implementing the Miura transformation, and that also generates animations for each, is given in Listing (4).

Listing 4: Maple program that transforms a mKdV equation solutionn to a KdV equation solutionn

```
# Miura transformation of mKdV equation to KdV equation
with(plots):
alias(u=u(x,t)): alias(v=v(x,t)):
# mKdV equation
pde_mKdV:=diff(v,t)-6*v^2*diff(v,x) + diff(v,x,x,x)=0;
# KdV equation
pde_KdV:=diff(u,t)-6*u*diff(u,x) + diff(u,x,x,x)=0;
# Known mKdv solution
sol1:=v=-k*tanh(k*(x+x0+2*k^2*t));
# mKdV plot data
dat1:={k=0.5,x0=0};
# Plot results
# =====
```




Figure 6: Initial profile at $t = -20$ of KdV equation *hump soliton* solution, $u = k^2 \left(2 \tanh [k(x + x_0 + 2k^2t)]^2 - 1 \right)$ for $k = 1$ and $x_0 = 0$. Solution move right to left.

```

z1:=subs(dat1,rhs(sol1)); # Set data values
# Animate solution
animate(z1,x=-50..10,t=0..80,
  numpoints=300,frames=50, axes=framed,
  labels=["x","v"],thickness=3,
  title="mKdV equation - kink soliton",
  labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
  titlefont=[TIMES, ROMAN, 16]);
# Check mKdV solution
pdetest(sol1,pde_mKdV);
# Define Miura transformation
MiuraTrans:=u=v^2+diff(v,x);
# Generate KdV solution using Miura transformation
sol2:=simplify(eval(subs(sol1,MiuraTrans)),size);
# Check KdV solution
pdetest(sol2,pde_KdV);
# KdV plot data
dat2:={k=0.5,x0=0};
# Plot results
# =====
z:=subs(dat2,rhs(sol2)); # Set data values
# Animate solution
animate(z,x=-50..10,t=0..80,
  numpoints=300,frames=50, axes=framed,
  labels=["x","u"],thickness=3,
  title="KdV equation - hump soliton",
  labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
  titlefont=[TIMES, ROMAN, 16]);

```

2.4 Bäcklund transformation of the KdV equation

We will now discuss a more convenient transformation than the Miura transformation due to Wahlquist and Estabrook [Wah-73].

Because the KdV equation is *Galilean invariant* and, following Drazin and Johnson [Dra-92], we decide to work with $u - \lambda$ rather than u . This has the effect of modifying the Miura transformation of eqn. (4) to

$$u = \lambda + v^2 + v_x, \quad (31)$$

where $u = u(x, t)$, $v = v(x, t)$ and λ is a real parameter. Thus, the mKdV equation (28) becomes

$$v_t - 6(v^2 + \lambda)v_x + v_{xxx} = 0. \quad (32)$$

Now eqns (31) and (32) are an alternative Bäcklund transformation as they imply KdV equation (25) for u . Clearly, we observe that if v is a solution to eqn. (32) then so is $-v$. From this observation, Wahlquist and Estabrook were led to introduce the two functions

$$u_1 = \lambda + v^2 + v_x, \quad u_2 = \lambda + v^2 - v_x \quad (33)$$

where λ and v are given. By adding and subtracting these equations we obtain

$$u_1 - u_2 = 2v_x, \quad u_1 + u_2 = 2(\lambda + v^2). \quad (34)$$

We now introduce an additional useful transformation,

$$u_i = \frac{\partial w_i}{\partial x}, \quad w = w(x, t), \quad (i = 1, 2), \quad (35)$$

from which we obtain, using eqns (33) and (34),

$$w_1 - w_2 = 2v \quad (36)$$

and

$$(w_1 + w_2)_x = 2\lambda + \frac{1}{2}(w_1 - w_2)^2. \quad (37)$$

Equation (36) is obtained by a single integration that requires the addition of an arbitrary function in $f(t)$. However, this arbitrary function can be absorbed into the definition of w_i without changing u_i and, therefore, there is no loss of generality by setting this function to zero. Equation (37) constitutes the x -part of the Bäcklund transformation for w_1 and w_2 which, along with eqn (35), enable solutions to the KdV equation to be generated. From eqns (32)-(36) and some algebraic manipulation we obtain the following t -part of the Bäcklund transformation,

$$(w_1 - w_2)_t - 3(w_{1x}^2 - w_{2x}^2) + (w_1 - w_2)_{xxx} = 0. \quad (38)$$

A Maple program that generates eqns (37) and (38) from the KdV equation by application of the Wahlquist and Estabrook transformation, is given in Listing (5).

Listing 5: Maple program that performs a Wahlquist and Estabrook transformation on the KdV equation

```
# Wahlquist and Estabrook transformation of mKdV equation
# to KdV equation
restart;with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t),v=v(x,t)): alias(w[1]=w[1](x,t),w[2]=w[2](x,t)):
# KdV and mKdV equations
pde_KdV:=diff(u,t)-6*u*diff(u,x) + diff(u,x,x,x)=0;
pde_mKdV:=diff(v,t)-6*v^2*diff(v,x) + diff(v,x,x,x)=0;
# Wahlquist and Estabrook mKdV equation
pde_mKdV_WE:=diff(v,t)-6*(v^2+lambda)*(diff(v,x)) +
diff(v,x,x,x)=0;
# Define transformations
# Miura
MiuraTrans:=u=v^2+diff(v,x);
#Wahlquist and Estabrook
WE_Trans:=isolate(subs(u=u-lambda,MiuraTrans),u);
# Derive new mKdV solution and check correct
mKdV_WE:=eval(subs(WE_Trans,pde_KdV));
mKdV_CK:=2*v*pde_mKdV_WE + diff(pde_mKdV_WE,x);
testeq(simplify(lhs(mKdV_WE)-lhs(mKdV_CK),size));
```

```

# Note: u and -u are both solutions of mKdV eqn
eqn1:=subs(u=u[1],WE_Trans);
eqn2:=subs({u=u[2],v=-v},WE_Trans);#unassign('u');
# Add and subtract WE transformations
eqn3:=eqn1-eqn2; eqn4:=eqn1+eqn2;
# substitute u=diff(w,x)
eqn5:=subs(u[1]=diff(w[1],x),u[2]=diff(w[2],x),eqn3);
eqn6:=subs(u[1]=diff(w[1],x),u[2]=diff(w[2],x),eqn4);
eqn7:=int(lhs(eqn5),x)=int(rhs(eqn5),x);
eqn8:=subs(isolate(eqn7,v),eqn6);
eqn9:=subs(isolate(eqn7,v),pde_mKdV_WE);
eqn10:=rhs(eqn8)/2=lhs(eqn8)/2;
# Final result
eqn11:=simplify(subs(eqn10,eqn9),size);

```

The Wahlquist and Estabrook transformation is relatively straight forward to use; for example, if we set $w_2 = 0$, eqn (37) becomes

$$w_{1x} = 2\lambda + \frac{1}{2}w_1^2, \quad (39)$$

which on integration yields

$$w_1 = -2k \tanh[kx + f(t)] \quad (40)$$

where we have set $\lambda = -k^2$ and f is an arbitrary function. Similarly, eqn (38) becomes

$$w_{1t} - 3w_{1x}^2 + w_{1xxx} = 0 \quad (41)$$

and, using eqn (39) twice to give $w_{1xxx} = w_{1x}^2 + w_1^2 w_{1x}$ we obtain

$$w_1 t - 2w_{1x} \left(w_{1x} + \frac{1}{2}w_1^2 \right) = 0. \quad (42)$$

Then, using eqn (39) once more for simplification, we can write eqn (41) as

$$w_{1t} + 4k^2 w_{1x} = 0. \quad (43)$$

This represents an *advection* equation with velocity $4k^2$, and which has the following general solution

$$w_1 = g(x, t) = g(x - 4k^2 t). \quad (44)$$

Now, for consistency with eqn. (40), the function $f(t)$ must take the form

$$f(t) = -4k^3 t - kx_0, \quad (45)$$

where x_0 is an arbitrary constant. The Bäcklund transformation therefore yields,

$$w_1 = -2k \tanh \left[k(x - x_0 - 4k^2 t) \right]. \quad (46)$$

Finally, we obtain from eqn (35)

$$u_1 = -2k^2 \operatorname{sech}^2 \left[k(x - x_0 - 4k^2 t) \right], \quad (47)$$

which is the same single soliton solution to the KdV equation, eqn (30), that we obtained using the Miura transformation and that is plotted in Fig(6). This solution is valid for $|w_1| < 2k$. However, for $|w_1| > 2k$ we obtain the singular solution,

$$w_1 = -2k^2 \operatorname{coth} \left[k(x - x_0 - 4k^2 t) \right], \quad (48)$$

and

$$u_1 = -2k^2 \left(1 - \operatorname{coth}^2 \left[k(x - x_0 - 4k^2 t) \right] \right). \quad (49)$$

A Maple program that derives the above example by application of the Wahlquist and Estabrook transformation, is given in Listing (6).

Listing 6: Maple program that performs a Wahlquist and Estabrook transformation on a KdV equation solution to obtain a mKdV equation solution

```

# Wahlquist and Estabrook transformation of KdV solution
# to mKdV solution
restart;with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t),v=v(x,t)): alias(w[1]=w[1](x,t),w[2]=w[2](x,t)):
# KdV and mKdV equations
pde_KdV:=diff(u,t)-6*u*diff(u,x) + diff(u,x,x,x)=0;
pde_mKdV:=diff(v,t)-6*v^2*diff(v,x) + diff(v,x,x,x)=0;
# Wahlquist and Estabrook mKdV equation
pde_mKdV_WE:=diff(v, t)-6*(v^2+lambda)*(diff(v, x))+diff(v, x, x, x)=0;
# x-part of BT
eqn1 := diff(w[1], x)+diff(w[2], x) = 2*((1/2)*w[1]-(1/2)*w[2])^2+
        2*lambda;
#t-part of BT
eqn2 := (1/2)*(diff(w[1], t))-(1/2)*(diff(w[2], t)) -
        3*(diff(w[1], x))^2*(1/2) +
        3*(diff(w[2], x))^2*(1/2) +
        (1/2)*(diff(w[1], x, x, x)) -
        (1/2)*(diff(w[2], x, x, x)) = 0;
# Set w[2]=0 in x-part of BT
eqn3:=eval(subs({w[2]=0},eqn1));
# Solve ODE and rearrange
eqn4:=eval(int(1/((w1^2+4*lambda)/2),w1)=int(1,x)+f(t));
eqn5:=isolate(simplify(subs(lambda=-k^2,eqn4),symbolic),w1);
# Set w[2]=0 in t-part of BT and cross multiply
eqn6:=eval(subs({w[2]=0},eqn2));
eqn6a:=numer(lhs(eqn6))*denom(rhs(eqn6))=numer(rhs(eqn6))*
        denom(lhs(eqn6));
# Simplify again by substitution
eqn7:=eval(diff(eqn3,x));
eqn7a:=lhs(eqn7)=subs(eqn3,rhs(eqn7));
eqn7b:=eval(diff(eqn7a,x));
# Simplify again by substitution
eqn8:=isolate(subs(eqn7b,eqn6),diff(w[1],t));
eqn8a:=simplify(eqn8,size);
# Simplify again by substitution
eqn9:=subs(eqn3,eqn8a);
eqn9a:=subs((1/2)*w[1]^2+2*lambda=diff(w[1],x),eqn9);
eqn9b:=subs(lambda=-k^2,eqn9a);
eqn9c:=lhs(eqn9b)- rhs(eqn9b) = 0;
# Solve advection pde and test
sol:=pdsolve(eqn9c);pdetest(sol,eqn9c);
# Solve for f(t)
eqn10:=k*x+k*f(t)=c*(-x+4*k^2*t)/(4*k^2);
f1:=solve(subs(c=-4*k^3,eqn10),f(t));
# Final solution for mKdV equation
eqn11:=subs(f(t)=f1,eqn5);
# Final solution for KdV equation
eqn12:=u1=simplify(diff(rhs(eqn11),x),trig);

```

3 Nonlinear superposition

3.1 Bianchi's Permutability Theorem

We need to first introduce the *Bianchi Permutability Theorem* that provides a simplified method of obtaining solutions to evolutionary equations by means of *nonlinear superposition*.

Theorem:[Con-08, 101]

Given a nonlinear PDE $F(u, x, t) = 0$ and its auto-Bäcklund transformation $BT(u, U; \lambda) = 0$, if one applies this BT to a given solution u_{n-1} with two different spectral parameters, $BT(u_{n-1}, u_n; \lambda_n) = 0$ and $BT(u_{n-1}, \tilde{u}_n; \lambda_{n+1}) = 0$, then there exists a fourth solution u_{n+1} which can be obtained by either $BT(u_n, u_{n+1}; \lambda_{n+1}) = 0$ or $BT(\tilde{u}_n, u_{n+1}; \lambda_n) = 0$, i.e. permuting the two spectral parameters.

[Put simply: from three distinct separate solutions to a given PDE, a fourth solution may be constructed by purely algebraic means using a suitable auto-Bäcklund transformation, if one exists].

This theorem is represented diagrammatically in figure 7.




Figure 7: Diagrammatic representation of Bianchi's theorem of permutability [Dra-92, p115].

No general proof of this theorem is known to exist and therefore it has to be proved for each PDE admitting an auto-Bäcklund transformation. Therefore, we will illustrate its use by examples.

3.2 Nonlinear superposition applied to the KdV equation

We have now developed the tools to enable a more straightforward method of obtaining solutions to the KdV equation using nonlinear superposition. From the above Bäcklund transformation we generate two solutions from the same seed solution but using two different values of λ . Thus, the x -part of the Bäcklund transformation, i.e. eqn. (37), can be written as

$$\begin{aligned} (w_1 + w_0)_x &= 2\lambda_1 + \frac{1}{2} (w_1 - w_0)^2, \\ (w_2 + w_0)_x &= 2\lambda_2 + \frac{1}{2} (w_2 - w_0)^2. \end{aligned} \quad (50)$$

Similarly, we construct two additional solutions: w_{12} from w_1 and λ_1 ; and w_{21} from w_2 and λ_2

$$\begin{aligned} (w_{12} + w_1)_x &= 2\lambda_1 + \frac{1}{2} (w_{12} - w_1)^2, \\ (w_{21} + w_2)_x &= 2\lambda_2 + \frac{1}{2} (w_{21} - w_2)^2. \end{aligned} \quad (51)$$

Now, the above Bianchi theorem of permutability states that w_{12} and w_{21} in eqns (51) will be equal, i.e.

$$w_{12} = w_{21}. \quad (52)$$

This provides a way of obtaining additional solutions to the KdV equation in a straightforward manner. First, subtract the difference of eqns (50) from the difference of eqns (51) to give

$$0 = 4(\lambda_2 - \lambda_1) + \frac{1}{2} [(w_{12} - w_1)^2 - (w_{21} - w_2)^2 - (w_1 - w_0) + (w_2 - w_0)^2]. \quad (53)$$

Then, using eqn (52), we obtain after some rearranging,

$$w_{12} = w_0 - \frac{4(\lambda_1 - \lambda_2)}{w_1 - w_2}. \quad (54)$$

Thus, we have arrived at the point where, given three solutions to the mKdV equation (28): w_0 , w_1 and w_2 , we can derive a fourth solution to the mKdV equation by purely algebraic means. Consequently, the mKdV solutions can be used to obtain corresponding KdV solutions by application of eqn (35).

A Maple program that derives the above result is given in Listing (7).

Listing 7: Maple program for the nonlinear superposition of KdV equation

```
# KdV Nonlinear Superposition
restart;
alias(w[0]=w[0](x,t)): alias(w[1]=w[1](x,t)):
alias(w[2]=w[2](x,t)):
alias(w[12]=w[12](x,t)): alias(w[21]=w[21](x,t)):
# Define two different x-part auto-Backlund transformations
eqn1:=diff((w[1]+w[0]),x)=2*lambda[1] + (1/2)*(w[1]-w[0])^2;
eqn2:=diff((w[2]+w[0]),x)=2*lambda[2] + (1/2)*(w[2]-w[0])^2;
# Define two additional different x-part auto-Backlund
# transformations
eqn3:=diff((w[12]+w[1]),x)=2*lambda[2] + (1/2)*(w[12]-w[1])^2;
eqn4:=diff((w[21]+w[2]),x)=2*lambda[1] + (1/2)*(w[21]-w[2])^2;
# Subtract difference of second BTs from first BTs
eqn5:=(eqn1-eqn2)-(eqn3-eqn4);
# Apply "integrability conditions"
w[21]:=w[12];
# Rearrange to obtain final solution
eqn6:=collect(isolate(eqn5,w[12]),w[0]);
```

As an example application of the above, let $w_0 = 0$ and use solutions from eqns (46) and (48), i.e. $w_1 = -2k \tanh([k(x - x_0 - 4k^2t)])$, $w_2 = -2k^2 \coth[k(x - x_0 - 4k^2t)]$; then eqn (54) yields,

$$w_{12} = \frac{-2(k_1^2 - k_2^2)}{k_1 \tanh(k_1(x - 4k_1^2t)) - k_2 \coth(k_2(x - 4k_2^2t))}, \quad (55)$$

where we have set the arbitrary constant $x_0 = 0$ (recall that $\lambda = -k^2$). Then finally from eqn (35) we obtain,

$$u_{12} = \frac{2(k_1^2 - k_2^2) [k_1^2 \operatorname{sech}^2(k_1(x - 4k_1^2t)) + k_2^2 \operatorname{csch}^2(k_2(x - 4k_2^2t))]}{(k_1 \tanh(k_1(x - 4k_1^2t)) - k_2 \coth(k_2(x - 4k_2^2t)))^2}, \quad (56)$$

which represents a two-soliton solution to the KdV equation. A plot of this solution is given in Fig (8).

A Maple program that derives the above example and includes an animation, is given in Listing (8).




Figure 8: Initial profile at $t = -4$ of KdV equation *2-soliton* solution,

$$u_{12} = \frac{2(k_1^2 - k_2^2) [k_1^2 \operatorname{sech}^2(k_1(x - 4k_1^2 t)) + k_2^2 \operatorname{csch}^2(k_2(x - 4k_2^2 t))]}{(k_1 \tanh(k_1(x - 4k_1^2 t)) - k_2 \coth(k_2(x - 4k_2^2 t)))^2}$$
 for $k_1 = 1/2$ and $k_2 = 1$.
 The solitons move left to right with the taller faster soliton overtaking the shorter slower soliton.

Listing 8: Maple program that generates a 2-soliton solution to the KdV equation by nonlinear superposition of mKdV equation solutions

```
# KdV/mKdV Equations, 2-soliton solution - Backlund Transformation
restart;with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t)): alias(w=w(x,t)): alias(v=v(x,t)):
# Define KdV and mKdV equations
pde_KdV:=diff(u,t)-6*u*diff(u,x) + diff(u,x,x,x)=0;
pde_mKdV:=diff(v,t)-6*v^2*diff(v,x) + diff(v,x,x,x)=0;
# Wahlquist and Estabrook transformation
pde1:=(w)->diff(w,x)=2*lambda+w^2/2;
# Define nonlinear superposition
w[12]:=w[0]-4*(lambda[1]-lambda[2])/(w[1]-w[2]);
# Define three mKdV solutions
w[0]:=0;
w[1]:=-2*k[1]*tanh(k[1]*(x-4*k[1]^2*t));
w[2]:=-2*k[2]*coth(k[2]*(x-4*k[2]^2*t));
# Calculate corresponding KdV solutions and test they are correct
sol1:=u=simplify(diff(w[1],x),trig);
sol2:=u=simplify(diff(w[2],x),trig);
pdetest(sol1,pde_KdV);
pdetest(sol2,pde_KdV);
# Solve for lambda using Wahlquist and Estabrook transformation
eqn1:=pde1(w[1]);
eqn2:=pde1(w[2]);
lambda[1]:=solve(eqn1,lambda);
lambda[2]:=solve(eqn2,lambda);
# Final solution from nonlinear superposition
eval(w[12]);
sol3:=u=diff(w[12],x);
# Check solution statisfies KdV equation
pdetest(sol3,pde_KdV);
# Apply manual simplification and check it is correct
sol3a := u = (2*(k[1]^2-k[2]^2))*(k[1]^2*sech(k[1]*(x-4*k[1]^2*t))^2+
             k[2]^2*csch(k[2]*(x-4*k[2]^2*t))^2)/
             (k[1]*tanh(k[1]*(x-4*k[1]^2*t))-
             k[2]*coth(k[2]*(x-4*k[2]^2*t)))^2;
testeq(rhs(sol3)=rhs(sol3a));
# Plot results
# =====
```



```

dat:={k[1]=1/2,k[2]=1};
u12:=subs(dat,rhs(sol3)); # Set data values
# Animate solution
animate(u12,x=-30..30,t=-4..4,
  numpoints=300,frames=50, axes=framed,
  labels=["x","u[12]"],thickness=3,
  title="KdV equation\n2-soliton solution",
  labelfont=[TIMES, ROMAN, 16],axesfont=[TIMES, ROMAN, 16],
  titlefont=[TIMES, ROMAN, 16]);

```

3.3 Three-soliton solution to the KdV equation using nonlinear superposition

The two-soliton solution presented above can be extended to generate multi-soliton solutions where, for a three-soliton solution, eqn. (54) becomes.

$$w_{123} = w_1 - \frac{4(\lambda_2 - \lambda_3)}{w_{12} - w_{13}}. \quad (57)$$

Or, on expansion,

$$w_{123} = \frac{\lambda_1 w_1 (w_2 - w_3) + \lambda_2 w_2 (w_3 - w_1) + \lambda_3 w_3 (w_1 - w_2)}{\lambda_1 (w_2 - w_3) + \lambda_2 (w_3 - w_1) + \lambda_3 (w_1 - w_2)}. \quad (58)$$

For example, if we define the following solutions to the mKdV equation,

$$w_0 = 0, \quad (59)$$

$$w_1 = -2 \tanh(k_1(x - 4k_1^2 t)), \quad (60)$$

$$w_2 = -2 \coth(k_2(x - 4k_2^2 t)), \quad (61)$$

$$w_3 = -2 \tanh(k_3(x - 4k_3^2 t)), \quad (62)$$

then, following the same procedure as for the two soliton solution, we obtain

$$w_{12} = \frac{-2(k_1^2 - k_2^2)}{k_1 \tanh(k_1(x - 4k_1^2 t)) - k_2 \coth(k_2(x - 4k_2^2 t))}, \quad (63)$$

$$w_{13} = \frac{-2(k_1^2 - k_3^2)}{k_1 \tanh(k_1(x - 4k_1^2 t)) - k_3 \tanh(k_3(x - 4k_3^2 t))}. \quad (64)$$

and from eqn (57) ,

$$w_{123} = \frac{N_1}{D_1}, \quad (65)$$

where,

$$\begin{aligned}
N_1 = & 2 \tanh(4tk_1^3 - xk_1) k_1^3 \coth(4tk_2^3 - xk_2) k_2 \\
& - 2 \tanh(4tk_1^3 - xk_1) k_1 \coth(4tk_2^3 - xk_2) k_2^3 \\
& - 2 \tanh(4tk_1^3 - xk_1) k_1^3 \tanh(4tk_3^3 - xk_3) k_3 \\
& + 2 \tanh(4tk_1^3 - xk_1) k_1 \tanh(4tk_3^3 - xk_3) k_3^3 \\
& + 2k_2^3 \coth(4tk_2^3 - xk_2) \tanh(4tk_3^3 - xk_3) k_3 \\
& - 2k_2 \coth(4tk_2^3 - xk_2) \tanh(4tk_3^3 - xk_3) k_3^3
\end{aligned} \quad (66)$$

and

$$\begin{aligned}
D_1 = & -\tanh(4tk_1^3 - xk_1) k_1 k_2^2 + \tanh(4tk_1^3 - xk_1) k_1 k_3^2 \\
& + \coth(4tk_2^3 - xk_2) k_2 k_1^2 - \coth(4tk_2^3 - xk_2) k_2 k_3^2 \\
& - \tanh(4tk_3^3 - xk_3) k_3 k_1^2 + \tanh(4tk_3^3 - xk_3) k_3 k_2^2.
\end{aligned} \tag{67}$$

The solution to the KdV equation is obtained by application of eqn (35) yielding,

$$u_{123} = \frac{N_2}{D_2}, \tag{68}$$

where,

$$\begin{aligned}
N_2 = & 2k_2^2 [k_1^2 (k_1 - k_2)^2 (k_1 + k_2)^2 \tanh^2(4tk_1^3 - xk_1) \\
& + 2 \tanh(4tk_3^3 - xk_3) k_1 k_3 (k_2 - k_3) (k_2 + k_3) (k_1 - k_2) (k_1 + k_2) \\
& \times \tanh(4tk_1^3 - xk_1) + k_3^2 (k_2 - k_3)^2 (k_2 + k_3)^2 \tanh^2(4tk_3^3 - xk_3) \\
& - (k_1 - k_3)^2 (k_1 + k_3)^2 (k_1^2 + k_3^2 - k_2^2)] \coth^2(4tk_2^3 - xk_2) \\
& - 4k_1 (k_1 + k_3) (k_1 - k_3) k_2 [(-k_2^2 k_1 + k_1^3) \tanh(4tk_3^3 - xk_3) \\
& \times \tanh^2(4tk_1^3 - xk_1) + k_3 (\tanh(4tk_3^3 - xk_3) - 1) \\
& \times (\tanh(4tk_3^3 - xk_3) + 1) (k_2 - k_3) (k_2 + k_3) \tanh(4tk_1^3 - xk_1) \\
& + (-k_1^3 + k_2^2 k_1) \tanh(4tk_3^3 - xk_3)] k_3 \coth(4tk_2^3 - xk_2) \\
& + 2k_1^2 [k_3^2 (k_1 - k_3)^2 (k_1 + k_3)^2 \tanh^2(4tk_3^3 - xk_3) \\
& + (k_2 - k_3)^2 (k_2 + k_3)^2 (-k_2^2 + k_1^2 - k_3^2)] \tanh^2(4tk_1^3 - xk_1) \\
& - 4k_3 \tanh(4tk_3^3 - xk_3) k_2^2 k_1 (k_2 - k_3) (k_2 + k_3) (k_1 - k_2) (k_1 + k_2) \\
& \times \tanh(4tk_1^3 - xk_1) - 2k_3^2 \tanh^2(4tk_3^3 - xk_3) \\
& \times (k_1 - k_2)^2 (k_1 + k_2)^2 (k_1^2 + k_2^2 - k_3^2)
\end{aligned} \tag{69}$$

and

$$\begin{aligned}
D_2 = & [(-k_3 k_1^2 + k_3 k_2^2) \tanh(4tk_3^3 - xk_3) + (-k_2^2 + k_3^2) \\
& \times \tanh(4tk_1^3 - xk_1) k_1 + k_2 \coth(4tk_2^3 - xk_2) (k_1 - k_3) (k_1 + k_3)]^2.
\end{aligned} \tag{70}$$

This represents a three-soliton solution to the KdV equation and a plot of this solution is given in Fig (9).

A Maple program that derives the above result and includes an animation, is given in Listing (9).

Listing 9: Maple program that generates a 3-soliton solution to the KdV equation by nonlinear superposition of mKdV equation solutions

```

# KdV/mKdV Equations, 3-soliton solution - Backlund Transformation
restart;with(PDEtools): with(PolynomialTools):
with(plots):
alias(u=u(x,t)): alias(w=w(x,t)): alias(v=v(x,t)):
# Define KdV and mKdV equations
pde_KdV:=diff(u,t)-6*u*diff(u,x) + diff(u,x,x,x)=0;
pde_mKdV:=diff(v,t)-6*v^2*diff(v,x) + diff(v,x,x,x)=0;
# Wahlquist and Estabrook transformation
pde1:=(w)->diff(w,x)=2*lambda+w^2/2;
# Define nonlinear superposition
w[123]:=w[1]-4*(lambda[2]-lambda[3])/(w[12]-w[13]);
w[12]:=w[0]-4*(lambda[1]-lambda[2])/(w[1]-w[2]);

```




Figure 9: Initial profile at $t = -40$ of KdV equation 3 -soliton solution, eqn. (68). Solitons move left to right with the taller faster solitons overtaking the shorter slower solitons.

```

w[13]:=w[0]-4*(lambda[1]-lambda[3])/(w[1]-w[3]);
w[123]:=collect( numer(w[123]),{lambda[1],lambda[2],lambda[3]})/
collect( denom(w[123]),{lambda[1],lambda[2],lambda[3]});
# Define four mKdV solutions
w[0]:=0;
w[1]:=-2*k[1]*tanh(k[1]*(x-4*k[1]^2*t));
w[2]:=-2*k[2]*coth(k[2]*(x-4*k[2]^2*t));
w[3]:=-2*k[3]*tanh(k[3]*(x-4*k[3]^2*t));
w[12]:=eval(w[12]);
w[13]:=eval(w[13]);
w[123]:=simplify(w[123],size);
# Calculate corresponding KdV solutions and test they are correct
sol1:=u=simplify(diff(w[1],x),trig);
sol2:=u=simplify(diff(w[2],x),trig);
sol3:=u=simplify(diff(w[3],x),trig);
pdetest(sol1,pde_KdV);
pdetest(sol2,pde_KdV);
pdetest(sol3,pde_KdV);
# Solve for lambda using Wahlquist and Estabrook transformation
eqn1:=pde1(w[1]);
eqn2:=pde1(w[2]);
eqn3:=pde1(w[3]);
lambda[1]:=solve(eqn1,lambda);
lambda[2]:=solve(eqn2,lambda);
lambda[3]:=solve(eqn3,lambda);
w[123]:=eval(w[123]);
sol123:=u=simplify(diff(w[123],x),size);
# Plot results
# =====
dat:={k[1]=1/2,k[2]=3/4,k[3]=1};
u123:=simplify(subs(dat,rhs(sol123)),size); # Set data values
# Animate solution
animate(u123,x=-40..40,t=-8..8,
numpoints=300,frames=50,axes=framed,
labels=["x","u[123]"],thickness=3,
title="KdV equation\n3-soliton solution",
labelfont=[TIMES,ROMAN,16],axesfont=[TIMES,ROMAN,16],
titlefont=[TIMES,ROMAN,16]);

```

3.4 Nonlinear superposition applied to the sine-Gordon equation

We take a similar approach to that used for the KdV equation and start by rewriting the x -part of the auto-Bäcklund transformation given in eqns (16) as

$$\begin{aligned}(v_1 - v_0)_x &= 2\lambda_1 \sin\left(\frac{v_1 + v_0}{2}\right), \\ (v_2 - v_0)_x &= 2\lambda_2 \sin\left(\frac{v_2 + v_0}{2}\right),\end{aligned}\tag{71}$$

where $v = v(x, t)$. Similarly, we construct two additional solutions: v_{12} from v_1 and λ_1 ; and v_{21} from v_2 and λ_2

$$\begin{aligned}(v_{12} - v_1)_x &= 2\lambda_1 \sin\left(\frac{v_{12} + v_1}{2}\right), \\ (v_{21} - v_2)_x &= 2\lambda_2 \sin\left(\frac{v_{21} + v_2}{2}\right).\end{aligned}\tag{72}$$

This provides a way of obtaining additional solutions to the S-G equation in a straightforward manner. First, subtract the difference of eqns (71) from the difference of eqns (72) to give

$$\begin{aligned}0 &= -2 \left[\sin\left(\frac{v_0 - v_1}{2}\right) + \sin\left(\frac{v_2 - v_{12}}{2}\right) \right] \lambda_1 + \\ &2 \left[\sin\left(\frac{v_0 - v_2}{2}\right) + \sin\left(\frac{v_1 - v_{21}}{2}\right) \right] \lambda_2.\end{aligned}\tag{73}$$

Again, the above Bianchi theorem of permutability states that v_{12} and v_{21} in eqns (72) will be equal, i.e.

$$v_{12} = v_{21}.\tag{74}$$

Then, using eqn (74) and some standard trigonometrical identities³, we obtain after some rearranging,

$$v_{12} = v_0 - 4 \arctan \left[\left(\frac{\lambda_1 + \lambda_2}{\lambda_1 - \lambda_2} \right) \tan \left(\frac{v_1 - v_2}{4} \right) \right].\tag{75}$$

Thus, as for the KdV equation, we have arrived at the point where, given three solutions to the S-G equation: v_0 , v_1 and v_2 , we can derive a fourth by purely algebraic means.

A Maple program that derives the above result is given in listing (10).

Listing 10: Maple program that generates a fourth solution to the sine-Gordon equation by nonlinear superposition of three known solutions

```
# Sine-Gordon Nonlinear Superposition
restart;
alias(v[0]=v[0](x,t),v[1]=v[1](x,t),v[2]=v[2](x,t),u=u(x,t)):
alias(v[12]=v[12](x,t):alias(v[21]=v[21](x,t)):
# Define sine-Gordon equation
pde_SG:=diff(u,x,t)=sin(u);
# Define two different x-part auto-Backlund transformations
eqn1:=diff((v[1]+v[0]),x)=2*lambda[1]*sin((v[1]-v[0])/2);
eqn2:=diff((v[2]+v[0]),x)=2*lambda[2]*sin((v[2]-v[0])/2);
# Define two additional different x-part auto-Backlund transformations
eqn3:=diff((v[12]+v[1]),x)=2*lambda[2]*sin((v[12]-v[1])/2);
eqn4:=diff((v[21]+v[2]),x)=2*lambda[1]*sin((v[21]-v[2])/2);
```

³Identities used: A) $\sin \alpha + \sin \beta = 2 \sin \frac{\alpha + \beta}{2} \cos \frac{\alpha - \beta}{2}$, B) $\sin(\alpha + \beta) = \sin \alpha \cos \beta + \cos \alpha \sin \beta$.

```

# Subtract difference of second BTs from first BTs
eqn5:=(eqn1-eqn2)-(eqn3-eqn4);
# Apply "integrability condition"
v[21]:=v[12];
# Rearrange
eqn6:=collect(eqn5,{lambda[1],lambda[2]});
# Apply trig identity: sin(A+B)+sin(A-B)=2*sin((A+B)/2)*cos((A-B)/2)
eqn7:=algsubs(sin(-(1/2)*v[1]+(1/2)*v[0])+sin(-(1/2)*v[12]+(1/2)*v[2])=
    2*sin((-1/2)*v[1]+(1/2)*v[0]-(1/2)*v[12]+(1/2)*v[2])/2)*
    cos((-1/2)*v[1]+(1/2)*v[0]+(1/2)*v[12]-(1/2)*v[2])/2,eqn6);
# Apply trig identity: sin(A+B)+sin(A-B)=2*sin((A+B)/2)*cos((A-B)/2)
eqn8:=algsubs(sin(-(1/2)*v[2]+(1/2)*v[0])+sin(-(1/2)*v[12]+(1/2)*v[1])=
    2*sin((-1/2)*v[2]+(1/2)*v[0]-(1/2)*v[12]+(1/2)*v[1])/2)*
    cos((-1/2)*v[2]+(1/2)*v[0]+(1/2)*v[12]-(1/2)*v[1])/2,eqn7);
# Remove common factor cos(..)
eqn9:=0=op(3,rhs(factor(eqn8)));

# Apply trig identity: sin(A+B)=sin(A)cos(B)+cos(A)sin(B)
eqn10:=algsubs(sin(-(1/4)*v[2]+(1/4)*v[0]-(1/4)*v[12]+(1/4)*v[1])=
    sin((1/4)*v[0]-(1/4)*v[12])*cos(-(1/4)*v[2]+(1/4)*v[1])+
    cos((1/4)*v[0]-(1/4)*v[12])*sin(-(1/4)*v[2]+(1/4)*v[1]),eqn9);
# Apply trig identity: sin(A+B)=sin(A)cos(B)+cos(A)sin(B)
eqn11:=algsubs(sin(-(1/4)*v[1]+(1/4)*v[0]-(1/4)*v[12]+(1/4)*v[2])=
    sin((1/4)*v[0]-(1/4)*v[12])*cos(-(1/4)*v[1]+(1/4)*v[2])+
    cos((1/4)*v[0]-(1/4)*v[12])*sin(-(1/4)*v[1]+(1/4)*v[2]),eqn10);
# Collect like terms
eqn12:=simplify(collect(eqn11,{sin((1/4)*v[0]-(1/4)*v[12]),
    cos(-(1/4)*v[2]+(1/4)*v[1])}),size);
# Rearrange
eqn13:=op(1,rhs(eqn12))=-op(2,rhs(eqn12));
# Define common factor
divisor:=select(has,lhs(eqn13),cos)*select(has,rhs(eqn13),cos);
# Remove common factor
eqn14:=convert(eqn13/divisor,tan);
# Final solution
isolate(eqn14,v[12]);

```

As an example application of the above, let $v_0 = 0$, $v_1 = \arccos [2 \tanh^2 (a_1 x^2 + t/a_1) - 1]$ (*peakon* soliton) and $v_2 = 4 \arctan [\exp (a_2 x + t/a_2)]$ (*anti-kink* soliton); then eqn. (75) yields the following new solution to the sine-Gordon equation,

$$v_{12} = -4 \arctan \left[\frac{\lambda_1 + \lambda_2}{\lambda_1 - \lambda_2} \tan \left\{ \frac{1}{4} \arccos [2 \tanh^2 (a_2 x + t/a_2) - 1] - \arctan [\exp (a_1 x + t/a_1)] \right\} \right]. \quad (76)$$

A plot of this solution is given in Fig (10).

A Maple program that derives the above example and includes an animation, is given in Listing (11).

Listing 11: Maple program that generates a 2-soliton solution to the sine-Gordon equation by nonlinear superposition

```

# Sine-Gordon equation - nonlinear superposition
restart; with(plottools): with(plots):
alias(u=u(x,t)): alias(v=v(x,t)):
# Define sine-Gordon equation
pde_SG:=diff(u,x,t)-sin(u)=0;
# Define S-G nonlinear superposition equation
v[12] := v[0]-4*arctan(tan(-(1/4)*v[2]+(1/4)*v[1])*
    (lambda[2]+lambda[1])/(lambda[1]-lambda[2]));
# Define three S-G solutions

```




Figure 10: Initial profile at $t = -200$ of sine-Gordon equation 2-soliton solution, $v_{12} = -4 \arctan \left[\frac{\lambda_1 + \lambda_2}{\lambda_1 - \lambda_2} \tan \left\{ \frac{1}{4} \arccos [2 \tanh^2 (a_2 x + t/a_2) - 1] - \arctan [\exp (a_1 x + t/a_1)] \right\} \right]$ for $a_1 = 5$, $a_2 = 4$, $\lambda_1 = 1$ and $\lambda_2 = 2$. The solution moves from right to left with the *peakon* overtaking the *anti-kink* soliton.

```
v [0] :=0;
v [1] :=arccos (2*tanh (a2*x+t/a2)^2-1);
v [2] :=4*arctan (exp (a1*x+t/a1));
# Check solutions are correct
pdetest (u=v [1], pde_SG); pdetest (u=v [2], pde_SG);
# Evaluate solution
v [12] :=eval (v [12]);
# Plot results
# =====
dat:={a1=5, a2=4, lambda [1]=1, lambda [2]=2};
zz:=subs (dat, v [12]);
animate (zz, x=-15..15, t=-200..200, numpoints=1000, frames=50,
numpoints=300, frames=50, axes=framed,
labels=["x", "u [12]"], thickness=3,
title="Sin-Gordon 2-soliton solution\nPeakon overtakes anti-Kink",
labelfont=[TIMES, ROMAN, 16], axesfont=[TIMES, ROMAN, 16],
titlefont=[TIMES, ROMAN, 16]);
```

References

- [Bac-80] **Bäcklund, A. V.** (1880), Zür Theorie der Partiellen Differentialgleichungen erster Ordnung, *Math. Ann.*, **XVII**, 285-328.
- [Con-08] **Conte, R. and M. Musette** (2008), The Painlevé Handbook, Springer, Dordrecht, Netherlands.
- [Dra-92] **Drazin, P. G. and R. S. Johnson** (1992). *Solitons: an introduction*, Cambridge University Press, Cambridge.
- [Dod-82] **Dodd, R. K., J. C. Eilbeck, J. D. Gibbon and H. C. Morris** (1982), *Solitons and nonlinear wave equations*, Academic Press.
- [Miu-68a] **Miura, R. A.** (1968). Korteweg-de Vries Equation and Generalizations I. A remarkable explicit transformation. *J. Math. Phys.*, **9**, pp1202-4.

- [Miu-68b] **Miura, R. A., C. S. Gardner and M. D. Kruskal** (1968). Korteweg-de Vries Equation and generalizations II. Existence of conservation laws and constants of motion. *J. Math. Phys.*, **9**, pp1204-9.
- [Rog-82] **Rogers, C. and W. F. Shadwick** (1982), Bäcklund transformations and their applications, Academic Press, New York.
- [Wah-73] **Wahlquist, H. D. and F. B. Estabrook** (1973). Bäcklund transformation for solutions to the Korteweg-de Vries equation. *Phys. Rev. Lett.*, **31**,1386-90.
- [Waz-09] **Wazwaz, Abdul-Majid** (2009). *Partial Differential Equations and Solitary Waves Theory*, Springer-Verlag, Berlin.